EVLA Memo 75

Approximate formulas for the distance term in far and near field interferometry

Tim Cornwell, NRAO

tcornwel@nrao.edu

Introduction

In many applications in wide field and near field interferometry, we need to calculate approximations to distance terms. In this *Mathematica* notebook, I calculate first and second order approximations for both far and near field imaging. I rederive the well known formulas for far-field imaging, and the familiar results for small object near field imaging. There is one new result - a second order formula for near field imaging of large objects. This last application will be described in more detail in a later memo.

I have used *Mathematica* for this work to avoid much tedious (and for me error prone) hand work. The *Mathematica* commands are given here so that the document can serve as a starting point for other work. Ignore them if desired.

The role of distance in interferometry

Consider a single emitter and two sensors. We define the vectors of (ra) an emitter from the center of the source, (r1, r2) a sensor from the center of the array (Figure 1). Using these, we can express the vector (v1,v2) from a sensor to an emitter.

```
Unprotect[l, m, n, d, x1, y1, z1,
  x2, y2, z2, X1, Y1, Z1, X2, Y2, Z2, U, V, W];
Clear[l, m, n, d, x1, y1, z1, x2, y2, z2, X1,
  Y1, Z1, X2, Y2, Z2, U, V, W];
<< Calculus`VectorAnalysis`
{l, m, n, d, x1, y1, z1, x2, y2, z2 ∈ Reals};
ra = d {ξ l, ξ m, Sqrt[1 - (ξ l)^2 - (ξ m)^2]};
r1 = {x1, y1, z1};
r2 = {x2, y2, z2};
v1 = ra - r1;
v2 = ra - r2;
R1 = Sqrt[DotProduct[v1, v1]];
R2 = Sqrt[DotProduct[v2, v2]];
Show[Import[
   "/Users/tcornwel/Projects/nearfield/papers/distance.png"]];
```



The propagation of the electric field from the emitter to the two sensors can be described by a Green's function:

$$g_1 = \frac{e^{\frac{2\pi i R_1}{\lambda}}}{i \lambda R_1}; \; g_2 = \frac{-e^{-\frac{2\pi i R_2}{\lambda}}}{i \lambda R_2};$$

The correlation between the electric fields at a and b is therefore given by

```
G12 = g1 g2;
G12 /. ξ → 1
```

$$e^{\frac{2 i \pi \sqrt{(d l-\mathrm{x1})^2+(d m-\mathrm{y1})^2+\left(d \sqrt{-l^2-m^2+1}-\mathrm{z1}\right)^2}}{\lambda}-\frac{2 i \pi \sqrt{(d l-\mathrm{x2})^2+(d m-\mathrm{y2})^2+\left(d \sqrt{-l^2-m^2+1}-\mathrm{z2}\right)^2}}{\lambda}} \Big/ \left(\sqrt{(d l-\mathrm{x1})^2+(d m-\mathrm{y1})^2+\left(d \sqrt{-l^2-m^2+1}-\mathrm{z1}\right)^2} \sqrt{(d l-\mathrm{x2})^2+(d m-\mathrm{y2})^2+\left(d \sqrt{-l^2-m^2+1}-\mathrm{z2}\right)^2}\, \lambda^2\right)$$

where R_a is the (scalar) distance of the emitter from the sensor a. Thus the essence of interferometry is expressed in the difference between the distances: $R_a - R_b$. We want to calculate the difference in distances of a given emitter from two sensors.

```
S = R1 - R2
```

$$\sqrt{(d l \xi-\mathrm{x1})^2+(d m \xi-\mathrm{y1})^2+\left(d \sqrt{-l^2 \xi^2-m^2 \xi^2+1}-\mathrm{z1}\right)^2}-\sqrt{(d l \xi-\mathrm{x2})^2+(d m \xi-\mathrm{y2})^2+\left(d \sqrt{-l^2 \xi^2-m^2 \xi^2+1}-\mathrm{z2}\right)^2}$$

We will work with various approximations to the distance term, using Taylor series in the direction cosines. To do this we use a dummy variable in the expansion. The following are first and second order expansions:

```
Sa[p_] := Normal[Series[S, {ξ, 0, p}]] /. ξ → 1;
```

The first order term is:

```
Sa[1]
```

$$-\frac{d l \mathrm{x1}}{\sqrt{d^2-2 \mathrm{z1} d+\mathrm{x1}^2+\mathrm{y1}^2+\mathrm{z1}^2}}+\sqrt{d^2-2 \mathrm{z1} d+\mathrm{x1}^2+\mathrm{y1}^2+\mathrm{z1}^2}-\sqrt{d^2-2 \mathrm{z2} d+\mathrm{x2}^2+\mathrm{y2}^2+\mathrm{z2}^2}-\frac{d m \mathrm{y1}}{\sqrt{d^2-2 \mathrm{z1} d+\mathrm{x1}^2+\mathrm{y1}^2+\mathrm{z1}^2}}+\frac{d l \mathrm{x2}}{\sqrt{d^2-2 \mathrm{z2} d+\mathrm{x2}^2+\mathrm{y2}^2+\mathrm{z2}^2}}+\frac{d m \mathrm{y2}}{\sqrt{d^2-2 \mathrm{z2} d+\mathrm{x2}^2+\mathrm{y2}^2+\mathrm{z2}^2}}$$

The second order term is:

```
Sa[2]
```

$$\frac{-2\,d\,l\,\mathrm{x1}-2\,d\,m\,\mathrm{y1}}{2\sqrt{d^2-2\,\mathrm{z1}\,d+\mathrm{x1}^2+\mathrm{y1}^2+\mathrm{z1}^2}}+\frac{1}{2}\sqrt{d^2-2\,\mathrm{z1}\,d+\mathrm{x1}^2+\mathrm{y1}^2+\mathrm{z1}^2}\left(\frac{d\,\mathrm{z1}\,l^2+d\,m^2\,\mathrm{z1}}{d^2-2\,\mathrm{z1}\,d+\mathrm{x1}^2+\mathrm{y1}^2+\mathrm{z1}^2}-\frac{(-2\,d\,l\,\mathrm{x1}-2\,d\,m\,\mathrm{y1})^2}{4\,(d^2-2\,\mathrm{z1}\,d+\mathrm{x1}^2+\mathrm{y1}^2+\mathrm{z1}^2)^2}\right)-$$
$$\frac{1}{2}\sqrt{d^2-2\,\mathrm{z2}\,d+\mathrm{x2}^2+\mathrm{y2}^2+\mathrm{z2}^2}\left(\frac{d\,\mathrm{z2}\,l^2+d\,m^2\,\mathrm{z2}}{d^2-2\,\mathrm{z2}\,d+\mathrm{x2}^2+\mathrm{y2}^2+\mathrm{z2}^2}-\frac{(-2\,d\,l\,\mathrm{x2}-2\,d\,m\,\mathrm{y2})^2}{4\,(d^2-2\,\mathrm{z2}\,d+\mathrm{x2}^2+\mathrm{y2}^2+\mathrm{z2}^2)^2}\right)+$$
$$\sqrt{d^2-2\,\mathrm{z1}\,d+\mathrm{x1}^2+\mathrm{y1}^2+\mathrm{z1}^2}-\sqrt{d^2-2\,\mathrm{z2}\,d+\mathrm{x2}^2+\mathrm{y2}^2+\mathrm{z2}^2}-\frac{-2\,d\,l\,\mathrm{x2}-2\,d\,m\,\mathrm{y2}}{2\sqrt{d^2-2\,\mathrm{z2}\,d+\mathrm{x2}^2+\mathrm{y2}^2+\mathrm{z2}^2}}$$

To be able to calculate quickly, we need versions of these equations that are amenable to being used in a Fourier transform. Thus we would like to obtain equations of the form `ul+vm+wn+c`. There are two principal forms that we need to consider, far field and near field.

The far field distance term

For a source on the celestial sphere, the distance D can be taken to be infinity, and the direction cosines add quadratically to unity. In doing this, we have to take into account the on-line tracking which takes out the nominal difference between the two z coordinates.

```
Sff[p_] := {Collect[Limit[
     Normal[Series[S + z1 - z2, {ξ, 0, p}]] /. ξ → 1, {d → Infinity}] /.
    {x2 → -u + x1, y2 → -v + y1, z2 → -w + z1}, {u, v, w}]}
```

First order:

The first order term is appropriate for small field of view:

```
Sff[1]
```

$(-l\,u-m\,v)$

Second order:

The second order term describes the problems induced by the non-coplanar baselines:

```
Sff[2]
```

$(-l\,u - m\,v + \frac{1}{2}(l^2 + m^2)\,w)$

As shown by Cornwell, Golap, and Bhatnagar (2003), this form is amenable to filtering in w, followed by Fourier transform in u,v.

If the array is coplanar, we can eliminate w:

```
Splane = Collect[Sff[2] /. {w → A u + B v}, {u, v}]
```

$((\frac{1}{2}A(l^2 + m^2) - l)\,u + (\frac{1}{2}B(l^2 + m^2) - m)\,v)$

This is simplified to a Fourier transform by redefining the direction cosines:

```
Splane /. {1 / 2 A (l^2 + m^2) - l → -L, 1 / 2 B (l^2 + m^2) - m → -M}
```

$(-L\,u - M\,v)$

Third order:

We can demonstrate that only the second order expansion is needed:

```
Sff[3]-Sff[2]
```

(0)

The near field distance term

The topic of near field imaging via interferometry was investigated by Carter (1988, 1989). His results were limited to the case of a small source. We can derive his results by taking the first order term for the near field.

First order:

```
Snf[p_] := {Simplify[Sa[p] //. {d^2 - 2 d z1 + x1^2 + y1^2 + z1^2 → d1^2,
     d^2 + x2^2 + y2^2 + z2^2 - 2 d z2 → d2^2}, {d1 > 0, d2 > 0}]}
Collect[Snf[1], {l, m}]
```

$$\left\{\mathrm{d1} - \mathrm{d2} + l\left(\frac{d\,\mathrm{x2}}{\mathrm{d2}} - \frac{d\,\mathrm{x1}}{\mathrm{d1}}\right) + m\left(\frac{d\,\mathrm{y2}}{\mathrm{d2}} - \frac{d\,\mathrm{y1}}{\mathrm{d1}}\right)\right\}$$

There are two ways to transform this equation to a simpler form. If we set up the array to be focused on the center of the target area then all of the square-root terms are equal to d.

```
toFocus = {d1 → d, d2 → d};
Simplify[Snf[1] //. toFocus, d > 0]
```

$\{l\,(\text{x2}-\text{x1})+m\,(\text{y2}-\text{y1})\}$

Thus we obtain a simple Fraunhofer term (leading to a Fourier transform). Alternatively, the sensor coordinates may be redefined by correcting for the extra distance to each sensor.

```
toSphere = {y1 → d1 Y1 / d, y2 → d2 Y2 / d,
   x1 → d1 X1 / d, x2 → d2 X2 / d, z1 → d1 Z1 / d, z2 → d2 Z2 / d};
Collect[Snf[1] //. toSphere //. {X2 → -U + X1, Y2 → -V + Y1}, {l, m}]
```

$\{\text{d1}-\text{d2}-l\,U-m\,V\}$

Thus again we obtain a Fourier transform, but with an additional differential focus term for the center of the target area.

Second order:

The second order term can be simplified in the same two ways: by focusing the array or by working in a modified coordinate system.

First, the equation for refocussing:

```
FullSimplify[(Snf[2]-Snf[1])//.toFocus//.z1->W+z2, {d>0, d1 d2 >
0}];
Collect[%,{l^2+m^2,l,m}]
```

$$\left\{\frac{(\text{x2}-\text{x1})(\text{x1}+\text{x2})\,l^2}{2\,d}+\frac{m\,((-\text{x1}-\text{x2})(\text{y1}-\text{y2})+(\text{x2}-\text{x1})(\text{y1}+\text{y2}))\,l}{2\,d}+\frac{1}{2}\,(l^2+m^2)\,W+\frac{m^2\,(\text{y2}-\text{y1})(\text{y1}+\text{y2})}{2\,d}\right\}$$

Now the equation for coordinate redefinition:

```
Snf2 =
 Collect[FullSimplify[(Snf[2] - Snf[1]) //. toSphere //. Z2 → -W + Z1,
   {d > 0, d1 d2 > 0}], {l^2 + m^2, l , m, W}]
```

$$\left\{\frac{(\text{d1}\,\text{X2}^2-\text{d2}\,\text{X1}^2)\,l^2}{2\,\text{d1}\,\text{d2}}+\frac{m\,(2\,\text{d1}\,\text{X2}\,\text{Y2}-2\,\text{d2}\,\text{X1}\,\text{Y1})\,l}{2\,\text{d1}\,\text{d2}}+\frac{1}{2}\,(l^2+m^2)\,W+\frac{m^2\,(\text{d1}\,\text{Y2}^2-\text{d2}\,\text{Y1}^2)}{2\,\text{d1}\,\text{d2}}\right\}$$

An alternate representation of this is the sum of two terms:

```
Snf2a1=(d2(m X1 - l Y1)^2 - d1(m X2 - l Y2)^2)/(2 d1 d2);
Snf2a2=(W+((d1 X2^2-d2 X1^2)+(d1 Y2^2-d2 Y1^2))/(d1
d2))(l^2+m^2)/2;
Snf2a=Snf2a1+Snf2a2
```

– *General::spell1 : Possible spelling error: new symbol name "Snf2a" is similar to existing symbol "Snf2".* More...

$$\frac{\text{d2}\,(m\,\text{X1} - l\,\text{Y1})^2 - \text{d1}\,(m\,\text{X2} - l\,\text{Y2})^2}{2\,\text{d1}\,\text{d2}} + \frac{1}{2}\,(l^2 + m^2)\left(W + \frac{-\text{d2}\,\text{X1}^2 + \text{d1}\,\text{X2}^2 - \text{d2}\,\text{Y1}^2 + \text{d1}\,\text{Y2}^2}{\text{d1}\,\text{d2}}\right)$$

To demonstrate that this is equivalent:

```
FullSimplify[Snf2-Snf2a]
```

{0}

The first **Snf2a1** is a shear term that couples together the x and y axes. This is quite troublesome! It originates in the distortion of the parabolic part of the distance term as seen from off axis.We can evaluate the ratio for typical values, and for the asymptotic case where the baseline length is much smaller than the distance to the source.

```
(Snf2a1/Snf2a2)//.{l->λ/R,m->λ/R,d1->d,d2->d,X1->B,X2->0,Y1->-B,Y2
->0,W->B}//.d->Infinity
```

0

Thus the shear term can normally be neglected, and so a good approximation to the second order distance term is:

```
Cancel[Collect[FullSimplify[Snf2a2+(Snf[1]//.toSphere//.{X2→
-U+X1,Y2→-V+Y1})],{l,m,l^2+m^2}]]
```

$$\left\{\text{d1} - \text{d2} - l\,U - m\,V + \frac{(l^2 + m^2)\,(-\text{d2}\,\text{X1}^2 + \text{d1}\,\text{X2}^2 - \text{d2}\,\text{Y1}^2 + \text{d1}\,\text{Y2}^2 + \text{d1}\,\text{d2}\,W)}{2\,\text{d2}\,\text{d1}}\right\}$$

This form is amenable to calculation - the first term is a constant per pair of sensors, and the rest may be performed by w projection.

Summary:

We have calculated first and second order approximations to the distance term in interferometry for both far and near field cases. These approximations have been found previously, except for the second order near field case for which we find a form allowing w projection to be used.

References:

Carter, W.H., (1988), Radio Science, **6**, 1085-1093.

Carter, W.H., (1989), IEEE-Trans. **AP-37**, 314-318.

Cornwell, T.J., and Perley, R.A., (1992), Astron.Astrophys., **261**, 353.

Cornwell, T.J., Golap, K., and Bhatnagar, S., (2003), EVLA memo 67.

Cornwell, T.J., Golap, K., and Bhatnagar, S., (2004), *in preparation..*